
ABSTRACT  
When comparing two populations, or modeling binary response, 
PROC TTEST is a valuable tool for evaluating numeric variables 
and an effective technique for identifying potential model variable 
candidates.  Once identified, the analyst can more quickly 
prioritize further investigation of these variables.  The 
disadvantage to PROC TTEST is that it does not summarize all 
the information you need into one easy to read report.  To get the 
whole story about a particular variable, you need to flip back and 
forth between three different pieces of output.  When there are 
literally hundreds or thousands of variables to evaluate, poring 
through the numerous parts and pages of PROC TTEST output 
can be a daunting, arduous task. 
 
This paper outlines code that takes the various parts of the PROC 
TTEST output and summarizes all of the vital information for each 
variable into a SAS data set.  The data set can then either be 
printed for use as a handy reference or exported to Excel for 
further cosmetic formatting.  The example presented in this paper 
will use Base SAS  and SAS/STAT and is appropriate for the 
beginning to intermediate statistical programmer or analyst. 

INTRODUCTION  
PROC TTEST is a useful tool for comparing two populations or for 
determining potential variables to use for modeling binary 
response.  However, the output from the following code is a bit 
cumbersome.  This code is examining the data set "file1" which is 
stored in the "mainlib" library.  The _numeric_ specification 
indicates that all numeric variables are to be tested. 
   proc ttest data = mainlib.file1; 
      var _numeric_; 
      class buy_ind; 
   run; 
The first section of output lists each variable as well as the 
number of observations falling into each by-group.  In this case, 
the class variable, buy_ind, has a value of 0 or 1 to indicate if a 
person is a non-buyer or a buyer, respectively.  The output also 
reports the mean and standard deviation of the buyers and non-
buyers for each variable.  

 
The question is, are the means of the by-groups for particular 
variables statistically different from each other?  For example, if 
you were looking at "Income" (var3), you would want to know if 
the mean, or average, income of buyers, $18,458, was statistically 
significantly different from the mean income of non-buyers, 
$23,150.   
 

 
To determine this, you must look at the second part of the output.  
The T-Tests section lists the results of testing the null hypothesis 
that the means of the two groups are equal under two different 
assumptions: 1: The variances of the two groups are equal and 2: 
The variances are not equal.   

 
As can be seen in the case of var3, the result of statistically 
different means is the same whether the variances are the same 
or not.  However, the results of the same tests for var8 yield 
slightly different results for the two variance equality assumptions.  
Thus, you must look to the third part of the output that tests the 

null hypothesis that the variances of the two groups are equal. 

Here it can be seen that the probability of the variances of the two 
groups being the same is very low and thus you should look at the 
T-test results under the assumption of unequal variances. 

When you have hundreds or thousands of variables to look at, 
this process can be tedious.  Manually turning this information into 
a report to use as a quick reference for your statistical modeling 
can also be very time consuming.  The remainder of this paper 
will outline steps that can be used to consolidate the output from 
PROC TTEST into a SAS data set that can be printed for use as 
a reference or output for further use as an appendix to a report. 
 
*SAS and all other SAS Institute Inc. product or service names are registered 
trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® 

THE RESULT 
The data set below is the end result of the code discussed in this 
paper.  It is a SAS data set that includes the variable name, the 
mean of the buyers, the mean of the non-buyers, the t-value, the 
probability and finally, a classification of the level of difference 
found between the two sample means based on an arbitrary cut-
off.  
 
Of course, the code can be modified to reflect any binary 
population and the cutoffs used to determine the level of 
difference between the two sample means can also be changed 
depending on the circumstances. 

 

THE PROCESS  
STEP ONE 
Using ODS, this section of code puts the PROC TTEST output 
into three different SAS data sets, one for each component of the 
output.  Macro variable references are utilized so that the code 
can be easily modified for testing additional data sets.   
 
The data set that is being tested is named "file1" and is stored in 
the "mainlib" library.  The resulting data sets to be created will be 
stored in the "mylib" library and the data set names will have 
"test1" as a prefix to easily identify them.  You may also consider 
making lib1 equal to the work directory to cut down on the number 
of permanent data sets being stored. 
   %let lib1 = mylib; 
   %let lib2 = mainlib; 
   %let fileorg = file1; 
   %let pre = test1; 
  
   ods output "Statistics" = &lib1..&pre.stats 
                    "T-Tests" = &lib1..&pre.ttests 
                    "Equality of Variances" = &lib1..&pre.vars; 
      proc ttest data=&lib2..&fileorg; 
         class buy_ind; 
         var _numeric_; 
      run; 
   ods output close; 
 
STEP TWO 
The following data set, mylib.test1stats, is the result of outputting 
the Statistics portion of the PROC TTEST through ODS.  This is 
the section of output containing the means and standard 
deviations of each class level of each variable. 

This section of code deals with that part of the PROC TTEST 
output.  The code makes it possible to have one row for each 
variable.  In the PROC TTEST output, there are three rows for 
each variable - one for buyers, one for non-buyers and one for the 
differences.  Each row in the resulting data set will have 2 new 
variables - mean_buyers and mean_nonbuyers.  These variables 
are equal to the variable that was previously called "mean".  
   %let lib1 = mylib; 
   %let pre = test1; 

data &lib1..&pre.stats1 (keep = variable class        
mean_nonbuyers mean_buyers); 

      set &lib1..&pre.stats (rename=(mean=avg)); 
      if class = '           0'  
        then mean_nonbuyers = avg; 
      if class = '           1' 
        then mean_buyers = avg; 
    run; 
 

 
At this point, mylib.test1stats1, there are still three rows for each 
variable.  This next piece of code separates the rows for buyers 
into one file and the rows for non-buyers into another file.  So, 
each of these two files will have only one row for each variable.  
Rows for differences are just dropped. 
   data &lib1..&pre.buyers(drop = class mean_nonbuyers)  
        &lib1..&pre.nonbuyers(drop = class mean_buyers); 
      set &lib1..&pre.stats1; 
      if class = '           0' then output &lib1..&pre.nonbuyers; 
      if class = '           1' then output &lib1..&pre.buyers; 
   run; 
 
Now the two separate files are sorted and merged back together 
on "Variable" and the only variables that remain are the means of 
the two groups and the variable name - one row for each variable.   
    proc sort data = &lib1..&pre.buyers; 
       by variable; 
   run;  
   proc sort data = &lib1..&pre.nonbuyers; 
      by variable; 
   run;    
   data &lib1..&pre.statsfinal; 
      merge &lib1..&pre.buyers 
         &lib1..&pre.nonbuyers; 
      by variable; 
   run; 
 
The final data set from this step, mylib.test1statsfinal, should 
look like this. 

 



 
STEP THREE 
This section deals with the second part of the PROC TTEST 
output - the part that tells you whether or not the means of the by-
groups for each variable are statistically equal or not.  In this 
section there are two lines for each variable. One showing the 
probability of the means being the same if the variances are the 
same, and one showing the probability of the means being the 
same if the variances are different.  
 
The following code creates a new variable, "difference", which 
classifies, into one of four categories, the strength of each 
probability of the means of the by-groups being different.  The 
code uses this data set, mylib.test1ttests, which was created 
above using ODS.   
 

   
   %let lib1 = mylib; 
   %let pre = test1; 
   data &lib1..&pre.ttests1; 
      set &lib1..&pre.ttests; 
      length difference $10.; 
      if probt le .0001 then difference = 'highly'; 
      else if probt le .005 then difference = 'somewhat'; 
      else if probt le .05 then difference = 'weak'; 
      else difference = 'not at all'; 
   run; 
    
The resulting data set, mylib.test1ttests1 should look like this. 

 
STEP FOUR 
The question now is, if any of the "difference" classifications for 
the same variable don't match, how to determine which probability 
of mean equality is the one to use without going through the last 
piece of output and manually checking the probability of the 
variances being the same for every single variable? The next 
piece of code merges the output that shows the probability of 
variances being equal to the output that shows the probabilities of 
means of the two by-groups being equal.  To avoid confusion with 
the "probf" variable in the means equality section, "probf" in the 
variance equality section is renamed as "probv".  
   data &lib1..&pre.vars; 
      set &lib1..&pre.vars (rename=(probf=probv)); 
   run; 
   proc sort data = &lib1..&pre.vars; 
      by variable; 
   run; 
   proc sort data = &lib1..&pre.ttests1; 
      by variable difference; 
   run; 
    data &lib1..&pre.merged; 

      merge &lib1..&pre.ttests1 (in=a) 
                &lib1..&pre.vars     (in=b keep = variable probv); 
      by variable; 
      if a; 
   run; 
  
The resulting data set, mylib.test1merged, will look like this.  
Note that the same value of "probv" from the variance equality 
output has been appended to both the "Equal" and "Unequal" 
rows of the mean equality output. 

 
Now, based on an arbitrary cutoff chosen for the value of the 
probability of the variances being equal, "probv", this code outputs 
only those rows where the value of "probv" matches the 'Equal' or 
'Unequal' indicator.  Thus, for example, if the probability of 
variance equality is <.0001, then the row where variances are 
'Unequal' will be output.  
   data &lib1..&pre.diffsame; 
      set &lib1..&pre.merged; 
      if probv le .0050 and variances = 'Unequal'  
         then output; 
      if probv gt .0050 and variances = 'Equal'  
         then output; 
   run; 

The resulting data set, mylib.test1diffsame, will look like this. 

STEP FIVE 
The final step merges the above file to the file with the means of 
the two groups.  The two files going in and final output file should 
all have just one row for each variable.  This final file will list the 
variable name, the mean of each by-group and the significance of 
the difference between those two means. 
   %let lib1 = mylib; 
   %let pre = test1; 
   proc sort data = &lib1..&pre.diffsame; 
      by variable; 
   run; 
   data &lib1..&pre.ttest; 
      merge &lib1..&pre.statsfinal   
      &lib1..&pre.diffsame (drop = method variances df probv); 
      by variable; 
   run; 
 
CONCLUSION  
This may look like a lot of code to do a simple task, especially 
when compared to the four lines of code in the introduction to this 
paper.  However, once you familiarize yourself with it, this code  
runs easily and will save hours of work in addition to producing an 
easy to use reference.  

